%% US Public Debt and Safe Asset Market Power
%% Jason Choi, Rishabh Kirpalani, and Diego Perez
%% Nov 24, 2024

%% Solve Monopoly Equilibrium

%----------------------------------------------------------------
% 0. Housekeeping
%----------------------------------------------------------------

close all

%----------------------------------------------------------------
% 1. Defining variables
%----------------------------------------------------------------

// Endogenous Variables
var b rb rkstar rk krw_star kus_star krw kus kstar k wstar w c_rw c_us drdb spread vrw vus dMrwdb y;

// Exogenous Variables
var nnu oomega A Astar;

// Shocks
varexo eps_nnu eps_oomega eps_A;

// Parameters
parameters ggamma bbeta eeta llambda aalpha Astarbar Abar iiota iiota_star ddelta_rw ddelta_us
  nnu_bar oomega_bar rrho_nnu rrho_oomega ssigma_nnu ssigma_oomega rrho_A ssigma_A rrho_Astar ssigma_Astar
  b_me rb_me rkstar_me rk_me krw_star_me kus_star_me krw_me kus_me kstar_me k_me
  capKstar_me capK_me wstar_me w_me crw_me cus_me spread_me vrw_me vus_me;

%----------------------------------------------------------------
% 2. Calibration
%----------------------------------------------------------------

% // Parameters
disp('% Epsilon = 2.2, Lambda = 2') 
ggamma = 2;
bbeta = 0.9886;
eeta = 0.545;
llambda = 2;
aalpha = 0.3;
Astarbar = 0.9254;
Abar = 0.8154;
iiota = 0.9070;
iiota_star = 0.7939;
ddelta_rw = 0.1;
ddelta_us = 0.1;
nnu_bar = 0.0041325;
oomega_bar = 0.0165;
rrho_nnu = 0.99;
ssigma_nnu = 0.01;
rrho_oomega = 0.95;
ssigma_oomega = 0.3;
rrho_A = 0.95;
ssigma_A = 0.01;
rrho_Astar = rrho_A;
ssigma_Astar = ssigma_A;

% Analytic Steady State (Monopoly Equilibrium)
b_me = (oomega_bar/(nnu_bar*eeta))^(1/(eeta-1-llambda));
rb_me = 1/bbeta - nnu_bar*(b_me)^(eeta-1) - 1;
rkstar_me = 1/bbeta + ddelta_rw - 1;
rk_me = 1/bbeta + ddelta_us - 1;
krw_star_me = ((aalpha*(1-iiota_star)*Astarbar*((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^((aalpha*(1-iiota_star)-1)/(aalpha*(1-iiota_star))))/(1/bbeta+ddelta_us-1))^((aalpha*(1-iiota_star))/(1-aalpha));
kus_star_me = ((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^(1/(aalpha*(1-iiota_star)))*krw_star_me^((1-iiota_star*aalpha)/(aalpha*(1-iiota_star)));
krw_me = ((aalpha*iiota*Abar*((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^((aalpha*iiota-1)/(aalpha*iiota)))/(1/bbeta+ddelta_rw-1))^((aalpha*iiota)/(1-aalpha));
kus_me = ((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^(1/(aalpha*iiota))*krw_me^((1-(1-iiota)*aalpha)/(aalpha*iiota));
kstar_me = krw_star_me + krw_me;
k_me = kus_star_me + kus_me;
capKstar_me = krw_star_me^iiota_star*kus_star_me^(1-iiota_star);
capK_me = krw_me^(1-iiota)*kus_me^iiota;
wstar_me = Astarbar*(1-aalpha)*(capKstar_me)^aalpha;
w_me = Abar*(1-aalpha)*(capK_me)^aalpha;
crw_me = wstar_me + (rkstar_me-ddelta_rw)*kstar_me + nnu_bar/eeta*(b_me)^eeta + rb_me*b_me;
cus_me = w_me + (rk_me-ddelta_us)*k_me - oomega_bar/(1+llambda)*(b_me)^(1+llambda) - rb_me*(b_me);
drdb_me = -nnu_bar*(eeta-1)*b_me^(eeta-2);
dMrwdb_me = 0;
nnu_me = nnu_bar;
oomega_me = oomega_bar;
A_me = Abar;
Astar_me = Astarbar;
spread_me = (rk_me-ddelta_us-rb_me);
vrw_me = crw_me^(1-ggamma)/(1-ggamma)/(1-bbeta);
vus_me = cus_me^(1-ggamma)/(1-ggamma)/(1-bbeta);
by_me = b_me/(A_me*(kus_me^iiota*krw_me^(1-iiota))^aalpha);
y_me = A_me*(kus_me^iiota*krw_me^(1-iiota))^aalpha;
cost_me = oomega_bar/(1+llambda)*(b_me)^(1+llambda);

%----------------------------------------------------------------
% 3. Model
%----------------------------------------------------------------

model;

c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(nnu*b^(eeta-1)+1+rb);
c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(1-ddelta_rw+rkstar);
c_rw + kstar + b = wstar + (1-ddelta_rw+rkstar(-1))*kstar(-1) + nnu(-1)/eeta*(b(-1))^eeta + (1+rb(-1))*b(-1);

c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(oomega*b^llambda+1+rb+drdb*b);
c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(1-ddelta_us+rk);
c_us + k - b = w + (1-ddelta_us+rk(-1))*k(-1) - oomega(-1)/(1+llambda)*(b(-1))^(1+llambda) - (1+rb(-1))*b(-1);

rk = Astar*aalpha*(1-iiota_star)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star)-1);
rkstar = Astar*aalpha*iiota_star*krw_star^(aalpha*iiota_star-1)*kus_star^(aalpha*(1-iiota_star));
rk = A*aalpha*iiota*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota-1);
rkstar = A*aalpha*(1-iiota)*krw^(aalpha*(1-iiota)-1)*kus^(aalpha*iiota);
wstar = Astar*(1-aalpha)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star));
w = A*(1-aalpha)*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota);

0 = -(drdb+nnu*(eeta-1)*b^(eeta-2))*(c_rw(+1)/c_rw)^(-ggamma)+(1+rb+nnu*b^(eeta-1))*dMrwdb;
0 = dMrwdb*(rkstar+1-ddelta_rw);

k = kus + kus_star;
kstar = krw + krw_star;

log(nnu) = (1-rrho_nnu)*log(nnu_bar) + rrho_nnu*log(nnu(-1)) + ssigma_nnu*eps_nnu;
log(oomega) = (1-rrho_oomega)*log(oomega_bar) + rrho_oomega*log(oomega(-1)) + ssigma_oomega*eps_oomega;
log(A) = (1-rrho_A)*log(Abar) + rrho_A*log(A(-1)) + ssigma_A*eps_A;
log(Astar) = (1-rrho_Astar)*log(Astarbar) + rrho_Astar*log(Astar(-1)) + ssigma_Astar*eps_A;

spread = (rk-ddelta_us-rb);

vrw = c_rw^(1-ggamma)/(1-ggamma) + bbeta*vrw(+1);
vus = c_us^(1-ggamma)/(1-ggamma) + bbeta*vus(+1);

y = A*(kus^iiota*krw^(1-iiota))^aalpha;

end;

%----------------------------------------------------------------
% 4. Computation
%----------------------------------------------------------------

initval;
  b = b_me;
  rb = rb_me;
  rkstar = rkstar_me;
  rk = rk_me;
  krw_star = krw_star_me;
  kus_star = kus_star_me;
  krw = krw_me;
  kus = kus_me;
  kstar = kstar_me;
  k = k_me;
  wstar = wstar_me;
  w = w_me;
  c_rw = crw_me;
  c_us = cus_me;
  drdb = drdb_me;
  dMrwdb = dMrwdb_me;
  nnu = nnu_me;
  oomega = oomega_me;
  spread = spread_me;
  A = A_me;
  Astar = Astar_me;
  vrw = vrw_me;
  vus = vus_me;
  y = y_me;
end;

resid;
check;

shocks;
    var eps_nnu = 1;
    var eps_oomega = 1;
    var eps_A = 1;
end;

set_dynare_seed('default');
stoch_simul(order=2,noprint,nograph,periods=100000,pruning);
% stoch_simul(order=1,noprint,nograph,periods=10000,pruning);

%----------------------------------------------------------------
% 5. Generate moments
%----------------------------------------------------------------

spread_path = (rk-ddelta_us-rb)*100;
var_sp = var(spread_path);
auto_sp = autocorr(spread_path);
cost = oomega./(1+llambda).*(b).^(1+llambda);
benefit = (eeta-1).*b.*(log(b)-nnu);
var_by = var(b./y);
auto_by = autocorr(b./y);
corr_pq_by = corr(spread_path,b./y);

moments = [mean(b./y) mean(spread_path) var_by var_sp corr_pq_by auto_by(2) auto_sp(2)]';
data_mom = [0.41 0.62 0.03 0.086 -0.56 0.96 0.70]';
rowNames = {'Mean b/y','Mean sp','Var b/y','Var sp','Corr (b/y,sp)','Autocorr b/y','Autocorr sp'};
colNames = {'Model Moments','Data Moments'};
TableA0 = array2table([moments data_mom],'RowNames',rowNames,'VariableNames',colNames)

deficit = cost(1:end-1) + b(2:end) - (1+rb(1:end-1)).*b(1:end-1);
deficit_y = deficit./y(1:end-1);
ca = -(b(2:end)-b(1:end-1)) + kus_star(2:end)-kus_star(1:end-1) - (krw(2:end)-krw(1:end-1));
ca_y = ca./y(1:end-1);
nfa = - b + kus_star - krw;
nfa_y = nfa./y;
var_ca_y = var(ca_y);
var_nfa_y = var(nfa_y);
var_deficit_y = var(deficit_y);
corr_nfa_y = corr(nfa_y,b./y);
corr_ca_def_y = corr(ca_y,deficit_y);

moments = [mean(rb) var(rb)*100 var_ca_y*100 var_nfa_y var_deficit_y corr_nfa_y corr_ca_def_y]';
data_mom = [0.0053 0.00097*100 0.00033*100 0.035 0.002 -0.654 -0.207]';
rowNames = {'Mean rb (Targeted)','Var rb(x100)','Var CA(x100)','Var NFA','Var Deficit','Corr(NFA,b)','Corr(CA,deficit)'};
colNames = {'Model Moments','Data Moments'};
TableA0 = array2table([moments data_mom],'RowNames',rowNames,'VariableNames',colNames)

nfa_pub_y = - b./y;
nfa_pvt_y = (kus_star - krw)./y;

var_nfa_pvt_y = var(nfa_pvt_y);
corr_nfa_pvt_y = corr(nfa_pvt_y,b./y);

nfa_noy = (- b + kus_star - krw)./y;
corr_nfa_b = corr(nfa_noy,b./y);

ratio_var_nfa_pvtnfa = var_nfa_pvt_y/var_nfa_y;

%----------------------------------------------------------------
% 6. Save moments for welfare calculations
%---------------------------------------------------------------

b_sme = mean(b);
rb_sme = mean(rb);
rkstar_sme = mean(rkstar);
rk_sme = mean(rk);
krw_star_sme = mean(krw_star);
kus_star_sme = mean(kus_star);
krw_sme = mean(krw);
kus_sme = mean(kus);
kstar_sme = mean(kstar);
k_sme = mean(k);
wstar_sme = mean(wstar);
w_sme = mean(w);
crw_sme = mean(c_rw);
cus_sme = mean(c_us);
drdb_sme = mean(drdb);
dMrwdb_sme = mean(dMrwdb);
nnu_sme = mean(nnu);
oomega_sme = mean(oomega);
A_sme = mean(A);
Astar_sme = mean(Astar);
vus_sme = mean(vus);
vrw_sme = mean(vrw);

%----------------------------------------------------------------
% 6. Calculate welfare from transition
%---------------------------------------------------------------'

b_me_lh = mean(b);
rb_me_lh = mean(rb);
spread_me_lh = mean(spread);

oo_me_lh = oo_;
M_me_lh = M_;
options_me_lh = options_;

save me_lh_save oo_me_lh M_me_lh options_me_lh b_me_lh spread_me_lh rb_me_lh;
